

Chapter 1

Matters Mathematical

1.1 Numerical Integration

Even if we know all the forces acting on a body, it is sometimes impossible to write down nice neat equations that describe the behavior of that body. In such cases we can often use a computer to simulate the behavior of the body. And sometimes, even if we can write the equations out, it is very instructive to do the simulation. Since we are going to spend a fair bit of time considering how the gravitational pull of body *A* influences the motion of body *B*, and vice versa, let us see how we might carry out such a simulation. We will let body *A* have a mass of 100 units and let it be located at the origin. Body *B* will have vanishingly small mass—we can let it equal 0. It will start off at a point 100 units out on the positive *x* axis, and it will be moving straight upwards at velocity V_y .

Two bodies are attracted to each other directly as the product of the masses and inversely as the square of the distance between them:

$$F = KM_1 \frac{M_2}{R^2}$$

where K is a constant that we will usually set to one. If the attracting body is at the origin and the body with which we are concerned is at point (x, y) , then the force in the x direction (F_x) is:

$$F_x = F \frac{X}{R} = -K \cdot M_1 \cdot M_2 \frac{X}{R^3},$$

and the force in the y direction (F_y) is:

$$F_y = F \frac{Y}{R} = -K \cdot M_1 \cdot M_2 \frac{Y}{R^3}.$$

From first year physics,

$$F = Ma,$$

or force equals mass times acceleration. Acceleration is the rate of change of velocity. If we accelerate for a time in DT increments, we will add to our current velocity an amount equal to the product of a and DT :

2 Chapter 1: Matters Mathematical




Fig. 1-1 Resolution of the force F into two components, F_x and F_y .

$$V = V + aDT.$$

We can carry this one step farther and write

$$X = X + V_x DT$$

$$Y = Y + V_y DT$$

where V_x and V_y are the velocities in the x and y directions and DT is again the size of the time step.

If we translate all this into BASIC and do it over and over, we will be causing the computer to “pretend to be like” (read: “simulate”) the real bodies:

```
Z = (X*X + Y*Y)
R = SQR(Z)
R^3 = Z^1.5 = ZZ
Ax = M1 X/ZZ
Ay = M1 Y/ZZ
Vx = Vx + AxDT
Vy = Vy + AyDT
X = X + VxDT
Y = Y + VyDT
```

One could print out x and y , and that could be the end of it; but it is much more interesting to see this displayed on the screen. We first put the computer in “graphics” mode and set the size ratio to 5/7. (An early dot matrix printer printed characters 5 dots wide and 7 dots high. Somehow it got into the BASIC graphics mode.) So we start by saying:

```
SCREEN 9
SR = 5/7
```

To plot a point on the screen, you have to tell the computer where to put it and what color you want it to be. Points on the screen are labeled 0–640 from left to right and 0–350 from top to bottom. Thus, if

Section 1.1: Numerical Integration 3

```
X0 = 320
```

and

```
Y0 = 175
```

then

```
PSET (X0, Y0),7
```

will put a white dot in the center of the screen. We normally have positive x extend to the right of the origin and positive y go up from the origin. Therefore, to plot a point at (x, y) we say

```
PSET(X0 + X, Y0 -Y*SR),7
```

Note that we add x and subtract y . To make the vertical scale the same as the horizontal scale, we multiply by the screen ratio (**SR**) which we have already set to 5/7.

When we use **PSET** to make a point a certain color, it stays that color until we change it or turn the computer off. That's what you want if you are going to print out the screen (see below, Section 1.1.1) but not if you are trying to watch the body as it whirls around; in that case, you want to do something like the following:

```
PSET(X0 + X, Y0-Y*SR),0  
X = X + Vx*DT  
Y = -Y + Vy*DT  
PSET(X0 + X, Y0 -Y*SR),7
```

First erase the old point (set its color to black), then move the body to its new x and y , then print the new position. Two **PSET** instructions close to each other sometimes interfere and you may need to stick in a delay loop such as:

```
FOR D = 1 TO 100:NEXT D
```

which just slows things down a little in order to make the **PSET** work. Since **PSET** leaves such a tiny dot, we are going to use

```
CIRCLE (X0 + X,Y0 - Y*SR),2,0
```

to make a bigger dot. Program **ONE** (page 5) shows the behavior of a central mass and a vanishingly small test mass.

The method shown above to calculate the new position is probably exactly what you would come up with yourself if you have ever studied calculus. And it is the method Leonard Euler came up with some 300 years ago. Since Euler is a well known and respected mathematician, you might think you were in good hands emulating him, but you would be wrong. The method is not "efficient," and that is a real sin to serious numerical analysts.

There is a reason for this aversion. From the time when Newton invented the calculus, about 1660, until computers were widely available, say 1960, you did numerical analysis by hand if you did it at all. The problem we have been looking over, the two body problem, can be solved in closed form, even when both bodies

4 Chapter 1: Matters Mathematical

have non-zero masses. That means that we can get equations that can be used to figure out where the two bodies will be at any future time if we know their starting positions, masses, and velocities. We don't have to go step by step but can jump right to the answer and, if you are facing thousands and thousands of iterations of the core steps, you really appreciate that closed form solution.

Add on one more body—make it three bodies all with non-zero masses—and closed form is out the window. There is no closed form solution to the three body problem. So it's back to numerical integration, and you are facing months and months of brain-rotting work to solve a single set of initial conditions by hand. You might be tempted to make DT (the time step) larger so you get done faster. The problem with that is that whatever method of numerical integration you select, the smaller the time step, the more accurate the results. Run program **ONE** and set $DT = 10$ and watch what happens to an elliptical orbit. So increasing DT to cut down on your work is not a viable option. (This concept is shown in Figures 1-2–1-5).

There are other much more efficient methods of numerical integration. "Fourth order Runge-Kutta" is one such. Although each step is more complicated than Euler's method, you can work with larger time steps and more than make up for the extra complication with increased speed. If you are faced with 6 months grinding labor and somebody offers to cut it in half, you jump at the chance. And you think somebody is a fool if he or she doesn't do the same.

But with computers we are talking six minutes rather than six months and most of that time is spent watching the machine do its thing. So run program **ONE** with initial $x = 100$ and gradually decrease V_y until the orbit doesn't close on itself. Then drop DT by a factor of two (to 0.5) and see the error decrease. With only two bodies, any orbit should repeat time after time. When it does not repeat, you are taking too large a time step. In fact, every time you run any numerical analysis or simulation program, if the results are in anyway surprising, try cutting DT in half and see if you get the same output.

1.1.1 Graphics

After you have written a program and debugged it and painted the results on the screen, you would probably like to print out a copy of what appears on the screen. There is a program that generally comes with the BASIC interpreter called **GRAPHICS**. To load it, at the **C:** prompt you type **GRAPHICS** and then **BASIC**. When you press the <Print Screen> button, it will print out an $8\frac{1}{2}'' \times 11''$ version of whatever is showing on the screen on the line printer. This is extremely convenient, but it has two unpleasant aspects.

If you have a circle on the screen, **GRAPHICS** will print it as a squashed ellipse. The people who wrote the **SCREEN 9** part of BASIC got the aspect ratio wrong so you have to use a screen ratio of $5/7$ if you want to draw circles on the screen. The people who wrote **GRAPHICS** made the mistake of getting the screen ratio right; at least, it is a mistake if you are trying to use **SCREEN 9** in BASIC. The upshot is that you have to decide before you run a program whether you want

Section 1.1: Numerical Integration 5

it to look right on the screen or on paper, and set $SR = 5/7$ for good screens and $SR = 1$ for good paper.

The other aspect of **GRAPHICS** that is extremely annoying is that after printing the screen out on the line printer, it does not return control to BASIC correctly. And, worse, the copy of your program is no longer correct. You have to press <Control-Alt-Delete> to reboot the computer. So what you have to do is:

1. Develop your program until you get it right with a proper picture on the screen.
2. Interrupt the program and save a copy of it on a disk. To exit the program, press <Ctrl-Break>, then click on Restart with the mouse.
3. Change the screen ratio to 1.
4. Rerun the program, getting a picture that is "too tall."
5. Press <Print Screen> and print out a copy of the screen.

With a newer machine, using Windows 95 or later, **GRAPHICS** doesn't work. But I can get to the MS-DOS prompt, and from there I can run BASIC. When I get a pretty picture on the screen that I would like to print, I press the <Print Screen> button and that makes a copy of the screen and puts it in the clipboard. I go to Windows and into the Paint program; I can then "paste" the contents of the clipboard into Paint. Once in Paint I can invert the colors so I get black on white, which saves my printer ribbon, and by stretching the vertical dimensions by 140% I can make circles be round once more. Then I can print the picture, and it comes out looking like many of the figures you see in this book.

Alternatively, you can change the program to black-on-white while you're still in BASIC. Normally, using **PAINT (0,0), 15** changes the background, and changing the **LINE** or **PSET** colors to 0 will produce a black and white display. (If the point (0,0) is already used, simply use another unused point.)

```
10 REM ONE
20 MM = 1
30 KEY OFF
40 SR = 5 / 7
50 DT = 1
60 X0 = 320
70 Y0 = 175
80 M = 100
90 CLS
100 INPUT "Distance of secondary from primary: <250> "; X
110 INPUT "Initial velocity in Y direction: <.35> "; VY
120 SCREEN 9
130 CLS
140 CIRCLE (X0, Y0), 7
150 PRINT "Initial X="; X; "      Initial Vy="; VY
160 LINE (0, 175)-(600, 175)
170 REM R is the distance between the two bodies
180 R = (X * X + Y * Y) ^ 1.5
190 REM acceleration is force over mass
200 FX = MM * M * X / R / MM
210 FY = MM * M * Y / R / MM
220 REM calculate the velocities
230 VX = VX - DT * FX'velocity =old velocity plus acceleration times time step
240 VY = VY - DT * FY
```

6 Chapter 1: Matters Mathematical

```

250 CIRCLE (X0 + X, Y0 - Y * SR), 2, 0 ` erase the old point
260 X = X + VX * DT'position = old position plus velocity times time step
270 Y = Y + VY * DT
280 CIRCLE (X0 + X, Y0 - Y * SR), 2, 7 ` plot the new point
290 FOR I = 1 TO 100: NEXT I
300 GOTO 170

10 REM ONEPRIME - for print out via GRAPHICS
20 REM TO DISPLAY SINGLE BODY MOVING ABOUT A GRAVITATING MASS
30 `
40 SR = 5 / 7
50 DT = 1
60 X0 = 320
70 Y0 = 175
80 M = 100
90 CLS
100 INPUT "Distance of secondary from primary: <250> "; X
110 INPUT "Initial velocity in Y direction: <.35> "; VY
120 SCREEN 9 `put computer in graphics mode
130 CLS
140 CIRCLE (X0, Y0), 7
150 PRINT "Initial X="; X; " Initial Vy="; VY
160 LINE (0, 175)-(600, 175)
170 REM Z is the distance between the two bodies
180 Z = (X * X + Y * Y) ^ .5: KY$ = INKEY$
190 REM acceleration is proportional to the force in each direction
192 ZZ = Z ^ 3
200 FX = M * X / ZZ
210 FY = M * Y / ZZ
220 REM calculate the velocities
230 VX = VX - DT * FX'velocity =old velocity plus acceleration times time step
240 VY = VY - DT * FY
250 X = X + VX * DT'position = old position plus velocity times time step
260 Y = Y + VY * DT
270 PSET (X0 + X, Y0 - Y * SR), 7'plot the new point
275 IF KY$ = CHR$(27) THEN END
280 GOTO 170

```

To make this discussion more concrete let's look at some examples. We will use the program called **EULER** (see below) which is an embodiment of Euler's method of calculating orbits. The first body is at the origin, is stationary, and has a mass of 100. Body 2 begins at $x, y = 200, 0$ and has a vertical velocity of 0.4.

```

10 REM EULER
20 KEY OFF
30 REM first body at 0,0 and stationary, mass=100
40 REM second body at 200, Vy=.4
50 X=200
60 Y=0
70 VX=0
80 VY=.4
90 M=100
100 DT=1
110 SR=5/7
120 X0=320
130 Y0=175
140 SCREEN 9
150 CLS: PAINT (0,0), 15
152 CIRCLE (X0,Y0),3,0
160 LOCATE 25,1
170 PRINT"Euler method dt="DT" single precision Vy="VY;
180 T0=TIMER
190 R=(X*X+Y*Y)^1.5
200 FX=100*X/R

```

Section 1.1: Numerical Integration 7




Fig. 1-2 One year of a planet on an elliptical orbit.




Fig. 1-3 One hundred fifty years of the planet shown in Figure 1-2. Note the "drift" caused by too large a time step.

```

210 FY=100*Y/R
220 VX=VX-FX*DT
230 VY=VY-FY*DT
240 X=X+VX*DT
250 Y=Y+VY*DT
330 PSET(X0+X,Y0-Y*SR)
340 IF Y>0 AND PY<0 THEN 400
350 PY=Y
360 GOTO 190
400 REM year's up
402 PY=Y
410 YR=YR+1
420 LOCATE 1,1
430 PRINT"Year="YR
440 IF YR<150 THEN 190
450 T1=TIMER-T0
460 PRINT "Time Rel to your CPU clock=",T1/60,"min"
470 END

```

8 Chapter 1: Matters Mathematical




Fig. 1-4 The same planet as shown in **Figure 1-3**, but taking time steps only half as large. Note the decreased “drift.”




Fig. 1-5 The same planet once again but showing that double precision arithmetic offers no improvement. Thus round-off error is not the problem.

Figure 1-2 shows a single orbit done with a time step of $DT = 1.0$; everything looks okay.

Figure 1-3 shows the trace of 150 orbits; you can see that the ellipses don't quite come back to the same place.

In Figure 1-4 we cut the time step by two ($DT = 0.5$) and most of the drift goes away.

Another bugaboo of numerical integration is round-off error, so in Figure 1-5 we go to double precision. There is some improvement, but the error caused by the finite time steps is quite a bit larger than the round-off error, so the former is the one about which we should worry.

1.2 Numerical Solution of an IVP

By Murray Schechter

The word “integration” is used several different ways in mathematics. The kind of integration sometimes called quadrature refers to the evaluation of a definite integral. That’s where the limit of a sum comes in. For instance, if someone were so unkind as to ask you to compute the integral from 1 to 2 of sine of X squared dx you’d have to resort to numerical integration. The method you learned in elementary calculus—namely, to find an antiderivative—won’t work here because there is no antiderivative that can be easily evaluated.

The word “integration” is also used in a related but distinctly different sense. It means to find the solution of an initial value problem (IVP). The Euler method of numerical integration uses the word in that sense. Suppose we seek a function y of t satisfying

$$\frac{dy}{dt} = f(t, y), y(t_0) = y_0,$$

(Here f is a given function of two variables and y_0 and t_0 are given numbers). Let the solution of this IVP be denoted by Y . Say you want a table or a graph of Y versus t for a certain range of t . This is not a problem of integration in the first sense; i.e., even if you could evaluate absolutely every definite integral that could be written down, it wouldn’t help you find the value of $Y(t_0 + 1)$, for instance. For some choices of the function f this IVP can be solved analytically, but for most it can’t. That’s where numerical analysis enters the picture.

Euler’s method is one of a large class of methods called “single step difference methods.” These all have the following form: pick a number h to serve as an increment for t and then approximate $Y(t + h)$ in terms of $Y(t)$. (Different methods use different ways of doing this approximation). $Y(t_0)$ is given, so you use it to approximate $Y(t_0 + h)$. Treating your approximation as though it really were $Y(t_0 + h)$, use it to approximate $Y(t_0 + 2h)$, and so on. In this way you can approximate $Y(t_0 + h)$, $Y(t_0 + 2h)$, $Y(t_0 + 3h)$, ... as far as you want to go, with the accuracy decreasing as you get further from t_0 . Here are the approximations used in some common methods:

Euler’s method:

$$Y(t + h) = Y(t) + h \cdot f(t, Y(t)).$$

Heun Method:

$$\text{let } k = f(t, Y(t)).$$

Then

$$Y(t + h) = Y(t) + h \frac{(k + f(t + h, Y(t) + hk))}{2}.$$



10 Chapter 1: Matters Mathematical

Fourth order Runge-Kutta method: let

$$k_1 = f(t, Y(t)),$$

$$k_2 = f\left(t + \frac{h}{2}, Y(t) + hk\frac{1}{2}\right),$$

$$k_3 = f\left(t + \frac{h}{2}, Y(t) + hk\frac{2}{2}\right),$$

$$k_4 = f(t + h, Y(t) + h \cdot k_3),$$

then

$$Y(t + h) = Y(t) + h \frac{(k_1 + 2k_2 + 2k_3 + k_4)}{6}.$$

The Euler method, while geometrically and analytically very easy to understand, is not very efficient at all, in the sense that other methods give a much more accurate approximation to the solution for the same amount of work. The Heun method, which is also easy to motivate geometrically, represents a considerable improvement, though it is far from state-of-the-art. The fourth order Runge-Kutta method represents a vast improvement and even though it is quite venerable and refinements of it exist, it's still a very respectable method. (Its derivation is purely analytical. At least as far as I know there is no way to come up with it by drawing a picture).

Let's demonstrate the difference in efficiency of these methods by taking a concrete example. Say f is the function given by $f(t, y) = \sin(ty)$, $t_0 = 0$ and $y_0 = 2$. (This IVP cannot be solved analytically, as far as I can tell). Let's try to find $Y(1)$ correct to 4 decimal digits. First we use Euler's method with $h = 1/8$, so we have to go 8 steps to approximate Y_1 . Then we'll do the computation all over again, starting from scratch with $h = 1/16$, going 16 steps to approximate Y_1 . We'll continue to halve h and see what happens to our approximation of Y_1 . Then we'll do the same with the Heun method and finally the fourth order R-K method. Some results are shown in Table 1-1.

Let's look at the last column first. The simplest rough and ready test of accuracy is that halving h doesn't seem to matter anymore. Using this criterion we can say that $Y_1 = 2.6697162$ to eight decimal digits. Since this criterion is not fool-proof, we could say with greater confidence that $Y_1 = 2.66972$ to six decimal digits. Euler's method is getting there, but at $h = 1/512$ it is correct only to four digits. (This means that it and the true value agree if both are rounded to four digits, but not if both are rounded to five digits.) The Heun method, on the other hand, is correct to five digits at $h = 1/128$. The R-K method is clearly the champion, giving 5 correct digits at $h = 1/8$. It's true that for a given h the Heun method requires about twice as much computation as the Euler method and the 4th order R-K method about four times as much. Still Euler comes out to be the least efficient by far.

h	Euler	Heun	R-K
$1/8$	2.6356774	2.6620215	2.6697116
$1/16$	2.6546952	2.6678357	2.6697160
$1/32$	2.6626766	2.6692524	2.6697162
$1/64$	2.6663107	2.6696011	2.6697162
$1/128$	2.6680416	2.6696876	
$1/256$	2.6688859	2.6697091	
$1/512$	2.6693028	2.6697144	

There are many more methods and classes than the three discussed above and much more to say about them. If you want to read all about it there are many good texts. You only have to read a few chapters to learn a lot about the subject, not the whole book. Two out of many good introductory texts are:

K. Atkinson, *Elementary Numerical Analysis* (New York: Wiley, 1985).

Yakowitz and Szidarovsky, *An Introduction to Numerical Computations* (New York: MacMillan, 1986).

You have to be reasonably comfortable with calculus to read these books.

You may have noticed that all these methods are for a single first order differential equation (i.e., only first order derivatives appear in the equation). In the next section we will see how to handle higher order systems.

1.2.1 Programs for the First Order IVP

The following are my versions of programs to do the first order IVP. I would say that they are loosely based on programs that Murray wrote, but he says he wants no part of anything as non-PC as a program in BASIC. C is much more PC than BASIC. Actually, you can't now, and never could, get less PC than BASIC.

—CCF

```

10 REM IVP
20 DEFDBL A-Z: KEY OFF
30 REM a Basic version of a program in C by Murray Schechter
40 CLS
50 PRINT "Euler method"
60 PRINT
70 INPUT "Initial T=      < 0 > "; IT
80 PRINT : INPUT "Final   T=      < 1 > "; FT
90 PRINT : INPUT "Initial Z=      < 2 > "; IZ
100 PRINT : INPUT "Number of steps= <2^n> "; NS
110 H = (FT - IT) / NS
120 REM body of program
130 z = IZ
140 BT = TIMER: T = 0
150 FOR I = 1 TO NS
160   z = z + H * SIN(T * z)
170   T = T + H
180 NEXT I
190 k = SIN(z * T): ET = TIMER

```

12 Chapter 1: Matters Mathematical

```
200 PRINT : PRINT "Euler="; z, ET - BT: PRINT
205 INPUT ans$: CLS
210 GOTO 40
```

For the Heun method, replace the word “Euler” by the word “Heun” and replace line 160 by:

```
160 K=SIN(T*Z)
162 Z=Z+H*(K+SIN((T+H)*(Z+H*K)))/2
```

For the R-K method replace the word “Euler” by “Runge-Kutta” and line 160 by:

```
160 T1=T+H/2
161 K1=SIN(T*Z)
162 K2=SIN(T1*(Z+H*K1/2))
163 K3=SIN(T1*(Z+H*K2/2))
164 K4=SIN((T+H)*(Z+H*K3))
165 Z=Z+(K1+2*K2+2*K3+K4)*H/6
```

1.3 Application of The Runge-Kutta Method to Orbit Calculation

By James Foster

Mention has been made of the Runge-Kutta (R-K) method of numerical integration of differential equations. The Euler method is used in programs **ONE**, **ONE-PRIME** and **EULER** above. R-K is capable of much greater precision than Euler, but the programming is substantially more complicated, and the execution time is correspondingly slower for a given number of integration steps.

The R-K solution of first order differential equations was discussed by Murray Schechter in the previous section. For orbit calculations, one needs to solve second order differential equations, and this section addresses that problem. Mr. Schechter has been very kind in explaining this to me, patiently answering my many questions. He wrote a detailed article describing and explaining this, and supplied a C language program for translation into BASIC.

In differential equations, one works with the idea of rates of change of a dependent variable with respect to a small change in the independent variable. One knows the formula for rate of change of the dependent variable, and tries to find its value. Many of these equations cannot be exactly solved by analytic methods, so a variety of numerical solutions have been developed. In particular the computation of orbits of three or more bodies requires numerical solutions. R-K is a widely used one.

We use the term “speed” or “velocity” (technically, “velocity” is a combination of speed and direction) to express the idea of a rate of change of position with respect to a change in time. The term “acceleration” is the rate of change of speed or velocity with respect to a change of time. Thus acceleration is the rate of change of the rate of change of position with respect to a change of time. This involves a second order differential equation, which must be solved to calculate an orbit.

Section 1.3: Application of The Runge-Kutta Method to Orbit Calculation 13

In orbit calculations, the acceleration of a body is determined by its position. This is what one is trying to calculate—thus one needs to know the answer in order to calculate the answer! Therefore one must make an estimation of the acceleration. In the Euler method, the assumption is that the acceleration of the body does not change during the time step. This of course is wrong, and is the cause of various errors that one sees in orbit displays. We are used to seeing a pseudo-precession in the two body display and high-speed ejections of one or more bodies when they approach each other closely. These effects are frequently caused by the simulation violating basic physical laws.

There are several kinds of R-K methods, but the most commonly used one is called “fourth order.” Here, one makes a series of four estimates of the acceleration and then calculates the position and velocity for that acceleration. The initial calculation is done at the start of the step interval of the independent variable. This is the same as in the Euler method. In R-K, one then makes two estimates of the acceleration at the halfway point of the step interval. In the first half-step estimate, the position predicted by the Euler method is used to calculate the acceleration and velocity. For the second half-step estimate the more refined estimate of the position (and consequent acceleration and velocity) from the last previous estimate is used. This might be thought of as “lifting oneself by the bootstraps.” Then a fourth estimate of the acceleration and velocity is made for the full step width. Here one uses the even more refined position calculated from the third step. Then a weighted average of the four accelerations and velocities is calculated. Intuitively, one would expect that the half-step calculations would be closer to the average values for the step width, and they are given twice the weighting of the estimates at the two ends of the step interval.

Finally, the weighted averages of the acceleration and velocity are used to calculate the best estimate of the new position and velocity of the body. Clearly, there must be errors, but it has been found that the error of this R-K method is on the order of being proportional to the fourth power of the step width. The Euler error, on the other hand, is on the order of being directly proportional to the step width. One can see that the R-K method is far superior for calculating very precise orbits.

The two body problem was exactly solved by Kepler. It is helpful, however, to apply numerical methods to the two body problem to explain numerical methods, because the programs are so short, and the accuracy of the numerical results can be checked by comparison with the exact solution. Kepler’s first law states that the two body orbit is an ellipse—it must end exactly where it started—and any deviation from this precisely measures the error of the numerical approximation. A demonstration program which uses these ideas follows. The user enters the eccentricity of the orbit and the number of steps to be used in calculating the orbit. The error is defined as the linear distance from the ending point of the orbit to the starting point. The error is then divided by the semi-major axis of the orbit to correct for different orbit sizes—I call this the relative error. The relative error is printed on the screen (the relative error is in exponential format because of the

14 Chapter 1: Matters Mathematical

Orbit Eccentricity	R-K Steps Per Orbit	Relative Error	Euler/R-K Time Ratio
0.1	100	4.4D-06	8.48
0.1	200	2.5D-07	18.35
0.1	400	1.4D-08	37.95
0.5	100	1.9D-04	1.72
0.5	200	9.4D-06	3.96
0.5	400	5.0D-07	8.53
0.5	800	2.9D-08	17.76
0.5	1600	1.7D-09	36.43
0.9	1000	8.8D-04	0.77
0.9	2000	3.1D-05	2.08
0.9	4000	1.2D-06	5.22
0.9	8000	5.7D-08	12.17
0.9	16000	3.0D-09	26.61
0.95	2500	2.9D-03	0.48
0.95	5000	9.8D-05	1.30
0.95	10000	3.4D-06	3.55
0.95	20000	1.3D-07	9.04
0.99	20000	7.3D-02	0.16

huge dynamic range of the relative errors.) For multiple superimposed orbits, there is little drift when the relative errors are less than about 1D-03 to 1D-04 (depending on how many orbits one wants to superimpose.) When one displays low eccentricity orbits using R-K, the small number of steps necessary for sufficient accuracy produces an unattractive appearance, since the displayed points are widely separated from each other.

In another much longer program I have compared the execution times of different orbits calculated with R-K and with the much larger number of steps needed by Euler to equal the error of R-K. Some sample results are found in Table 1-2.

Note that for low-precision work, such as for our display orbits on the screen and for highly eccentric orbits, the execution time for Euler may be comparable to R-K for equal accuracy. For “real world” astronomical orbit calculations, where the greater the accuracy the better, R-K is supreme. The “Euler” program I used for the comparison is not a true Euler method, but is the one used in the program **EULER**. The sequence of calculations is “wrong,” in that following the calculation of the estimated acceleration the estimated velocity is done prior to the estimated position. The correct sequence is acceleration followed by position and finally velocity. Amazingly, this “wrong” sequence produces much greater accuracy, and the errors are second order (errors approximately proportional to the square of the step width!)

Section 1.3: Application of The Runge-Kutta Method to Orbit Calculation 15

There is an analogous program which uses the polar coordinate system. It has only a single dependent variable—the radius—which changes as the independent variable—the central angle—increases. This program is shorter than one that uses the conventional rectangular coordinate system. The program using the polar coordinate system can calculate highly eccentric orbits with much greater precision for a given number of steps per orbit. I am greatly indebted to Mr. Schechter for providing the formula for the rate of change of the radius for a change in the central angle.

```
5 REM RK4
10 `Calculate an orbit using R-K. ORBITRK4.ASC
20 DEFDBL A-Z
22 KEY OFF
24 CLS
26 SF = 5 / 7
30 INPUT "Enter the Eccentricity of the orbit (0 - .95)"; EC
40 INPUT "Enter the number of steps per orbit (10 - 5000)"; NSTEPS%
50 CLS
60 SCREEN 9
70 XCENTER = 320: YCENTER = 175
80 CIRCLE (XCENTER, YCENTER), 3
90 M = 100#: R0 = 200# `M = mass, R0 = apoapsis distance
100 PRINT "Eccentricity ="; EC;
110 GOSUB 650 `CalcInitVals
120 T0 = 0: TF = PD
130 DT = (TF - T0) / NSTEPS%
140 X = R0: Y = 0#: VX = 0#: VY = V `Initial values
150 GOSUB 330 `RK
160 RKXERROR = R0 - X: RKYERROR = Y
170 RKERROR = SQR(RKXERROR ^ 2# + RKYERROR ^ 2#)
180 RKRELEERROR = RKERROR / A
190 PRINT "R-K number of steps ="; NSTEPS%
200 PRINT USING "Error / Semimajor axis =##.#^"; RKRELEERROR
210 PRINT "Want to see more orbits (Y/N)?"
220 A$ = INKEY$: IF A$ = "" THEN 220
222 IF A$ = "y" THEN 240
230 IF A$ <> "Y" THEN 320
240 X = R0: Y = 0#: VX = 0#: VY = V `Initial values
250 CLS : INPUT "Enter the number you want to see"; NUMORB%
260 CIRCLE (XCENTER, YCENTER), 3
270 FOR K% = 1 TO NUMORB%
280 GOSUB 330
290 LOCATE 2, 1: PRINT "Orbit #"; K%;
300 NEXT K%
320 END
325 REM *****
330 `RK:
340 `START OF RK -----
350 FOR J% = 1 TO NSTEPS%
360 EX = X: EY = Y: EVX = VX: EVY = VY `E = Estimated
370 GOSUB 720 `CalcEstAccels
380 EAX1 = EAX: EAY1 = EAY `Save the estimates `A = Acceleration
390 EVX1 = EVX: EVY1 = EVY `Save the estimates `V = Velocity
400 TMSTP = DT / 2# `Half-step estimates, TmStp = Time Step
410 GOSUB 780 `EstVals
415 GOSUB 720 `CalcEstAccels
420 EAX2 = EAX: EAY2 = EAY `Save'em
430 EVX2 = EVX: EVY2 = EVY `Save'em
440 `Boot-stap up
450 GOSUB 780 `EstVal
455 GOSUB 720 `CalcEstAccels
460 EAX3 = EAX: EAY3 = EAY `Save'em
```

16 Chapter 1: Matters Mathematical

```
470 EVX3 = EVX: EVY3 = EVY 'Save'em
480 'Now a full-step & more boot-strapping
490 TMSTP = DT
500 GOSUB 780 'EstVals
505 GOSUB 720 'CalcEstAccels
510 EAX4 = EAX: EAY4 = EAY 'Save'em
520 EVX4 = EVX: EVY4 = EVY 'Save'em
530 'Calculate weighted average accelerations
540 AVAX = (EAX1 + 2# * (EAX2 + EAX3) + EAX4) / 6#
550 AVAY = (EAY1 + 2# * (EAY2 + EAY3) + EAY4) / 6#
560 'Calculate weighted average velocities
570 AVVX = (EVX1 + 2# * (EVX2 + EVX3) + EVX4) / 6#
580 AVVY = (EVY1 + 2# * (EVY2 + EVY3) + EVY4) / 6#
590 'Final integrations
600 X = X + DT * AVVX: Y = Y + DT * AVVY
610 VX = VX + DT * AVAX: VY = VY + DT * AVAY
620 PSET (X + XCENTER, YCENTER - Y * SF)
630 NEXT J%
640 RETURN
645 REM *****
650 'CalcInitVals:
660 V = SQR(M * (1 - EC) / R0) 'Initial velocity
670 EN = (V * V / 2) - (M / R0) 'Specific mechanical energy
680 A = -M / (2 * EN) 'Semi-major axis
690 PI = 4# * ATN(1#)
700 PD = 2 * PI * A ^ 1.5 / SQR(M) 'Period
710 RETURN
715 REM *****
720 'CalcEstAccels:
730 'Inputs EX, EY & M; Outputs EAX & EAY
740 R2 = EX * EX + EY * EY 'Faster than EX ^ 2 + EY ^ 2
750 R = SQR(R2): R3 = R * R2 'Faster than R2 ^ 1.5
760 EAX = -M * EX / R3: EAY = -M * EY / R3
770 RETURN
775 REM *****
780 'EstVals:
790 'Inputs X, Y, VX, VY, EAX, EAY, EVX, EVY & TmStp
800 'Outputs EVX, EVY, EX & EY
810 'Must be done in the right order!
820 EX = X + TMSTP * EVX: EY = Y + TMSTP * EVY
830 EVX = VX + TMSTP * EAX: EVY = VY + TMSTP * EAY
840 RETURN
```

1.4 Centrifugal Force Concentration Point for a Sphere

By Richard McCusker

The gravitational force of attraction between two particles follows Newton's law, $Fg = G m_1 m_2 / R^2$, where the m 's are the masses of the particles, R is the distance between them, and G is the gravitational constant. For congregate masses, the literature commonly explains that, if the masses are spheres, the force acts as though the masses are concentrated at the centers of the respective spheres.

When a small mass is on a circular orbit around a large mass, the centrifugal force on the small mass is given by the equation

$$F_c = km\omega^2 R,$$

where ω is the angular speed, m is the mass, and R is the radius of the orbit. Over the years I have seen and used examples of centrifugal force where the masses

Section 1.4: Centrifugal Force Concentration Point for a Sphere 17




Fig. 1-6 Centrifugal force acting on a spherical body.

were always spherical and the orbital radii were always measured to the center of the sphere. However, I have never seen anything in the literature to confirm that this is valid. I presume that it is out there, but I have never run across it.

I just worked on the math aspect of a non-spherical mass in circular orbit, the object being to determine the point in the mass to which the orbiting radius would be measured. This undertaking did not include confirmation of the case of a spherical mass. So that became my current project.

The question was: does a homogeneous, rigid, spherical mass on a circular orbit act as though the centrifugal force on its mass were concentrated at its center? The answer turned out to be yes. The mathematical solution is given below.

In Figure 1-6 as oriented, due to symmetry, all of the vertical components of centrifugal force on the mass cancel, leaving only the horizontal to be summed. Also keep in mind that all particles of the mass have the same orbital speed ω , whether they lie in the plane of the paper or are above or below it.

F_{TC} is the centrifugal force on a small test mass m_T , and F_{TV} and F_{TH} are respectively the V and H components of it:

$$F_{TC} = km_T\omega^2\sqrt{(R-x)^2 + a^2}$$

$$F_{TH} = F_{TC}\cos\theta = km_T\omega^2(R-x).$$

This means that each mass particle of the sphere has an H component of centrifugal force on it proportional to its horizontal distance from the center of the principal mass.

Having established this, the H force equation for the incremental disk can be set up and solved:

$$dF_{CH} = k(dm)\omega^2(R-x),$$

18 Chapter 1: Matters Mathematical




Fig. 1-7 A sine wave.

where dm is the disk's incremental mass. Since the mass is equal to the density times the volume,

$$dm = \rho \pi y^2 dx.$$

And since $y^2 = b^2 - x^2$,

$$F_{CH} = k\rho\pi\omega^2 \int_{-b}^{+b} (R-x)(b^2-x^2)dx = k\left[\frac{4}{3}\pi b^3\rho\right]\omega^2 R.$$

The term in brackets is the mass of the sphere. Therefore $F_{CH} = km\omega^2 R$, proving that the orbital radius is measured to the center of the sphere where the mass could be considered concentrated. Q.E.D.

1.5 Weighing Planets

There are now more than a dozen known extra-solar planets, and the number seems to grow every week. As you might expect, given the youth of the techniques used to discover them, most of the ones found are quite large—one to ten times the size of Jupiter. Usually, they are reported as being “at least n times Jupiter” in mass. How do they know that and why do they not give a clean number rather than a minimum?

To understand that, we have to understand how they go about making those measurements. There are several methods of discovering extra-solar planets. We will examine only one here, that which involves the Doppler shift the planet induces in the spectrum of the primary body. If a star is moving away from the Earth, we see its light shifted toward the red end of the spectrum. If it is approaching Earth, the shift is toward the blue end. The light given off by a star is not a smooth, uniform spectrum. There are bright lines that are produced by elements at elevated temperatures. These are called *emission* lines. Similarly, cool elements absorb light and leave dark lines in the spectrum. Naturally enough, these are called *absorption* lines. The Doppler shift causes these lines, both emission and absorption, to move toward one end or the other of the spectrum.




Fig. 1-8 A planet revolving around a distant star.




Fig. 1-9 A planet on an elliptical orbit.

The would-be planet-discoverer picks a line in the star's spectrum and measures its frequency over time. If there is indeed a planet moving around the star there will be a shift in the frequency of the line that looks something like Figure 1-7.

Given such a graph, one can extract an large amount of data. First of all, there is the length of time that elapses between peaks—blue peak to blue peak or red peak to red peak. As the planet swings around the star the center of mass of the star-planet system stays fixed or in steady motion. Look then at Figure 1-8.

When the planet is moving toward the Earth, the star is moving away. When the planet returns to the same place in its orbit the star will be doing the same thing as it did the last time and the frequency of the observed line will be back to where it was last time. We have to remember that the Earth is moving around the Sun in a 12 month cycle. One group forgot to allow for this and "discovered" a planet with a one year period. Another one to watch out for is the effect of our own Ju-

20 Chapter 1: Matters Mathematical

piter in swinging around the Sun and the rest of our solar system—about an 11.8-year cycle.

After factoring out these sources of error we have the period of the new planet. We are examining stars that are a lot like our own Sun, so we can assume that they have a similar mass. Remembering that the period P is given by

$$P = k \left(\frac{R^3}{M} \right)^{0.5},$$

where M is the mass of the star, k is a constant that includes π and the gravitational constant, and R is the distance from the star to the planet. We can solve this for R , and, knowing the period and the constant, we know the radius of the new planet's orbit.

If the orbit of the new planet is edge-on toward the Earth then the star will be moving directly toward and away from the Earth at some point of its orbit. But if the orbit is face-on, the star will be jiggling up and down but not causing the frequency of the light it is emitting to change at all. The nearer to edge-on, the more effective the planet will be in modulating the star's spectrum. If we assume that the orbit is edge-on then we can figure out the smallest planet that could cause the star to move the way we see it move. If the orbit is tilted, the planet must be more massive in order to generate that much movement. That's why they say "at least."

If the planet is on a circular orbit, the modulation of the spectrum will be symmetric and sinusoidal. If, however, the orbit is elliptical, the modulation will have a somewhat different shape. The program below assumes that the planet is 1/100 the of the star's mass and that the line of sight toward the Earth goes off to the upper right at 45 degrees. The program draws the orbit and the consequent change in the frequency of the observed light. An initial value of "1" for the Y velocity will generate a circular orbit, and values less than 1 will generate ellipses. Remember that if you want a circular orbit to look circular on the screen you have to set the screen ratio to 5/7 in line 30.

```
10 REM DOPPLER
20 KEY OFF
30 SR = 5 / 7
40 DT = .25
50 X0 = 320
60 Y0 = 175
70 CLS
75 PRINT " Numbers between .1 and 1.5 generate reasonable display " \***
80 INPUT "Initial Y velocity of body 2 (1 makes circular):", VY2
90 VY1 = -VY2 / 99
100 PRINT : PRINT
110 M1 = 99
120 M2 = 1
130 X1 = -1
140 X2 = 99
150 SCREEN 9
160 CLS
170 CIRCLE (320, 175), 3, 15
180 LINE (0, 175)-(600, 175)
190 LINE (0, 300)-(600, 300)
200 LOCATE 2, 40
```

Section 1.6: Three Quickies 21

```
210 PRINT "Looking at velocity from 45 degrees"
220 REM r1 is distance from 1 to 2
225 ky$ = INKEY$: `
230 R1 = ((X1 - X2) * (X1 - X2) + (Y1 - Y2) * (Y1 - Y2)) ^ 1.5
240 REM do the velocities
250 VX1 = VX1 - (X1 - X2) * DT * M2 / R1
260 VY1 = VY1 - (Y1 - Y2) * DT * M2 / R1
270 VX2 = VX2 - (X2 - X1) * DT * M1 / R1
280 VY2 = VY2 - (Y2 - Y1) * DT * M1 / R1
290 X1 = X1 + VX1 * DT
300 Y1 = Y1 + VY1 * DT
310 X2 = X2 + VX2 * DT
320 Y2 = Y2 + VY2 * DT
330 PSET (X0 + X1, Y0 - Y1 * SR), 7
340 PSET (X0 + X2, Y0 - Y2 * SR), 3
350 VV = .707 * (VX1 + VY1)
360 T = T + .1
370 PSET (T, 300 + 2000 * VV), 2
375 IF ky$ = CHR$(27) THEN GOTO 385
380 GOTO 220
385 REM jump out of the loop:
395 END:
```

1.6 Three Quickies

By Lester Pecan

Here are three neat and short programs that Les submitted. Each one performs one or more useful functions and is only a few lines long.

—CCF

```
10 REM GAMMA
20 CLS
30 KEY OFF
40 PRINT "This program calculates the factor 'GAMMA' by which time and body
length are affected as a body moves at speeds approaching that of light.
Gamma is a function of the ratio R of body speed to light speed (186200
mi/sec)."
```

```
50 PRINT ""
60 INPUT "RATIO OF BODY SPEED TO SPEED OF LIGHT, 0-1 R"; R
70 G = (1 / (1 - R ^ 2)) ^ .5: REM This is the Lorentz transformation.
80 PRINT ""
90 PRINT ""
100 PRINT "GAMMA EQUALS: "; G
110 PRINT ""
120 PRINT ""
130 PRINT "Time is slowed (dilated) by a factor of"; G, "Dimensions along the
direction of travel shorten by a factor of"; 1 / G, "(In accord with Ein-
stein's Special Theory of Relativity, GAMMA is derived from the Lorentz
transformation.)"
```

```
10 REM ORBITALT
20 KEY OFF: CLS
30 ` ORBITALT calculates the required altitude above the earth's sur-
face for a satellite to achieve a desired orbital
period.
40 PRINT " "
50 R = 6372000!
60 INPUT "DESIRED SATELLITE PERIOD IN DAYS OR HOURS="; T
70 INPUT "If period entered is in HOURS, enter H ;
If period entered is in DAYS, enter D "; D$
80 IF D$ = "D" OR D$ = "d" THEN 120
```

22 Chapter 1: Matters Mathematical

```
90 A = (21618 * (3600 * T) ^ (2 / 3) - R) / 1000 * .6214
100 PRINT "
      SATELLITE ALTITUDE IS"; INT(A); "MILES FOR A PERIOD OF"; T; "HOURS"
110 END
120 T = T * 24
130 B = (21618 * (3600 * T) ^ (2 / 3) - R) / 1000 * .6214
140 Z = T / 24
150 PRINT "
      SATELLITE ALTITUDE IS"; INT(B); "MILES FOR A PERIOD OF"; Z; "DAY(S)"

10 REM SATPEROD
20 CLS : KEY OFF
30 PRINT "`SATPEROD' Calculates earth satellite normal (radial) accelera-
      tion,tangential "
32 PRINT "velocity & period as a function of its altitude, H, in miles above
      the"
34 PRINT "earth's surface.It also gives the escape velocity from any orbit."
35 PRINT
40 PRINT " It is based on Kepler's law: T^2=(4pi^2/G*M)*R^3, where
      T=period, "
42 PRINT "G=universal gravitational constant, M=earth's mass, and R=radius
      from earth "
44 PRINT "center to satellite."
50 PRINT " "
60 M = 5.98E+24: G = 6.67E-11: RE = 6372000!: REM RE=earth radius,meters
70 INPUT "SATELLITE ALTITUDE IN MILES, Height ="; H
72 PRINT
80 a = G * M / (RE + 1609 * H) ^ 2
90 RA = (RE + 1609 * H) * 6.214 * 10 ^ -4: PRINT USING "(Center-to-Center
      radius,miles, R=#####)"; RA
100 PRINT USING "Acceleration in meters/sec/sec= ##.####"; a
110 PRINT USING "Acceleration in G's= ##.####"; a / 9.8237
120 V = SQR((RE + 1609 * H) * a) * 2.237
130 PRINT USING "Velocity in miles per hour=#####"; V
140 VE = SQR(2) * V: PRINT USING "Escape velocity at this altitude,
      MPH,#####"; VE
150 PI = 3.14159265#
160 T = SQR((4 * PI ^ 2) * (RE + 1609 * H) / a) / 60
170 PRINT USING "Period in minutes= #####.# "; T
180 PRINT USING "Period in hours = ####.## "; T / 60
190 PRINT USING "Period in days = ##.####"; T / (60 * 24)
200 REM Since the moon has appreciable mass relative to earth, the center
210 REM of rotation of this binary system is their combined center of mass.
220 REM This point is about 2883 miles from earth center;thus, as a first
230 REM approximation, use as a lunar orbital radius the perigee/apogee mean
240 REM minus 2883: ((221463+252710)/2)-2883 = 234203 miles. This gives
250 REM a good approximation of the sidereal period of the moon.
260 END
```

1.7 Simple Harmonic Motion

When the force seeking to return a body to zero is linearly proportional to the displacement from that point, the body undergoes “simple harmonic motion” or SHM for short. If:

$$F = -k \cdot x,$$

$$a = \frac{F}{m},$$




Fig. 1-10 A simple pendulum.

$$y = \int (a \cdot dt),$$

$$x = \int (v \cdot dt),$$

or

$$\frac{d^2}{dt^2}x + \frac{k}{m}x = 0,$$

these conditions are met and x will be a sinusoidal function of time

$$x = A \sin(t + \phi),$$

where A and ϕ are arbitrary constants.

Note that a weight suspended from a spring and confined to motion in the vertical direction will undergo SHM. A pendulum, on the other hand, does not; as a matter of fact, a pendulum is not a very good time keeper, Mr. Galilei notwithstanding. To understand why, consider Figure 1-10, where

$$F = mg \cdot \sin(\text{Ang}).$$

If the restorative force were proportional to the displacement, we would have SHM, but it isn't. Now if Ang is very small it is approximately true that

$$\text{Ang} = \sin(\text{Ang}),$$

and the motion is almost sinusoidal.

Back a couple of hundred years ago, the most popular clock was a "bracket" or mantle clock with a short pendulum giving a beat of $\frac{1}{2}$ second. The short pendulum had to swing perhaps 30 degrees either side of vertical in order to work the escapement. Thirty degrees is 0.525 radians and the sine of 30 degrees is 0.500.




Fig. 1-11 A “fusee” used to make the “pull” of a spring more constant.

Thus the restorative force grows more slowly than the angle. As the spring winds down, the force it exerts on the gear train grows smaller, the pendulum swings in a shorter arc, and the clock runs faster—just the opposite of what you might expect.

There are several ways to combat this:

1. Make the pendulum longer so its swing can be shorter and its timekeeping better. This approach leads to “grandfather” clocks with pendulums that have a one second beat. Some of these tall case clocks were up to nine feet tall with $1\frac{1}{2}$ second pendulums.
2. Use falling weights rather than springs. These exert a constant force and the pendulum then swings a constant arc and the clock keeps quite good time—at least until it gets dirty and friction increases.
3. Use springs with a fusee (Figure 1-11). A fusee is a cone shaped device with a track to hold a chain. When wound up, the chain is mostly on the fusee. As the spring unwinds it pulls the chain off the fusee and winds it around the spring barrel. At the beginning, the fusee is of a small diameter but the spring is pulling at its maximum. As the chain is transferred to the spring barrel the fusee gets bigger in diameter so the lessening force that the spring exerts is balanced by the bigger lever arm due to the increased diameter of the fusee. If the taper of the fusee is properly chosen it matches the loss in power of the spring and the force applied to the movement of the clock stays constant without the height required to allow weights to drop.
4. Christiaan Huygens, 1629–1695 invented “cycloidal cheeks” (see Figure 1-12) which are fixed constraints that have the effect of making the pendulum get shorter as it swings to the side. This shortening is just right to compensate for the sinusoidal restorative force and result in true SHM.




Fig. 1-12 Huygen's method of making a short pendulum keep better time—"Huygen's cheeks."

1.8 Furlongs per Fortnight

Speaking of strange units like parsecs... my high school physics teacher was once asked whether he wanted the answers to some homework problems (that will give you a clue as to how old I am) in English or metric. His response was, "Use any units you like," but he wasn't best pleased when I reported velocities in furlongs per fortnight. To aid you in bedeviling the enemy, a furlong is 220 yards or 660 feet. A fortnight is 14 days (think about that one). Two useful numbers to remember are that 60 miles per hour is 88 feet per second and there are 86,400 seconds in a 24 hour day. If you work it all out, 60 MPH is equal to 161,280 furlongs per fortnight, or one foot per second is 1,832.7 furlongs per fortnight. Would those be FPFs?

—CCF

Measures of weight, distance, time and money can be bewilderingly complex. Four farthings to the penny, twelve pence to the shilling and twenty shillings to the pound. Would you prefer 12.5 cents to the bit and eight bits to the dollar? Perhaps, 14 pounds to the stone, 8 stones to the hundredweight (112 pounds) and 20 hundredweight to the ton (2240 pounds). How about 20 grains to the scruple, three scruples to the dram and 8 drams to the apothecaries' ounce. Don't forget that there are also troy ounces and avoirdupois ounces.

In Edinburgh, a friend and I were involved in some sort of calculation that involved conversion of units and I recited the old mnemonic "A pint's a pound the world around," meaning 16 ounces. To which Sid replied, "Not here, it bloody well isn't. Here its 20 ounces to the pint." Which explains, I think, why there are five quarters in an imperial gallon.

26 Chapter 1: Matters Mathematical

I have an old map with a distance scale marked in “German military miles” which is a reminder that every country had its own measures of distance and weight, and sometimes more than one of each. It is no wonder that rational people welcomed the metric system and were willing to give up their national units. When I was an undergraduate we often visited the Howard Atheneum, sometimes referred to as “The Old Howard,” on Saturday nights and after the presentation we would adjourn next door to Mr. Worth’s establishment where you could buy “seidels” of light or dark beer. It wasn’t until I was looking things up for this book that I discovered that a seidel was an Austrian unit of volume of 0.354 liters, or roughly 12 ounces. At least, a couple were enough for the three mile walk across the Charles to the dorms.

When the world adopted the metric system, it was supposed to adopt a whole set of prefixes for the units. They are:

deca	10	giga	10^9
hecto	100	tera	10^{12}
kilo	1000	peta	10^{15}
mega	1 million	exa	10^{18} .

Astronomers have not been on the cutting edge here. They have not only invented weird units of their own, but have persistently failed to use the proper prefixes. Take the astronomical unit, the distance from the Earth to the Sun. It is close to 149 million kilometers. Now the metric committee invented those prefixes so you wouldn’t have to express things in clumsy ways like “millions of kilo some-things.” You are supposed to say that the Earth is 149 gigameters from the Sun. Or that it is 0.149 terameters to the Sun. I haven’t heard either of those too often. As another example consider the light-year, a strange unit no matter how you look at it. A year contains 31 million seconds (another useful number to remember) and light goes 3×10^8 meters per second so a light-year is 93×10^{15} meters. We could round that up to 100×10^{15} or 10^{17} meters. Take 10 light-years, roughly 3 parsecs, and you have approximately one exameter. Who could possibly have trouble with that?

Speaking of units, a cubic foot of water weighs 62.5 pounds or exactly 1000 ounces. Is this a coincidence? Or does somebody know the story behind it? 12^3 equals 1728 cubic inches to the cubic foot, which doesn’t seem to be a big help.

1.9 Small Angles

The minute hand on our tall case clock is six inches long, and if you watch it very closely, you can see it move with each tick of the pendulum. The minute hand makes one revolution per hour thus turning 360 degrees in 3600 seconds, or one tenth of a degree (6 arc minutes) per second. Five such steps is the half degree diameter of the full Moon. So, for angles of about one degree a grandfather clock can give us examples. For smaller angles we have to look elsewhere.




Fig. 1-13 Two sine waves with wavelength L . The second wave lags behind the first by ϕ (ϕ).

A US dime (10 cent piece) has a diameter of 1.75 cm. At arms length (26 inches for me) it subtends 1.5 degrees. 26 inches is 66 cm. At small angles $\sin \text{Ang} = \tan(\text{Ang}) = \text{Ang}$ so that $1.75/66$ equals 0.0265 radians or 1.5 degrees. A dime at one meter will subtend an angle of 0.0175 radians or very nearly exactly 1 degree.

At 60 meters (roughly the distance from the goal posts to the 50 yard line of an American football field), a dime subtends 1 arc minute if you can even see it. Using a telescope or binoculars is not fair. At 3600 meters (roughly 2 miles) a dime subtends 1 arc second. As we all know, a parsec is the unit of distance such that the diameter of the Earth's orbit (two AUs) subtends an angle of one arc second.

1.10 A Primer on Resolution

Astronomers insist that bigger telescopes provide them with increased light-gathering power and improved resolution. The first is transparently true. The larger the objective, the more light it will intercept. Double the diameter and you multiply the light gathering potential by four. So whether you are using a photographic plate or a CCD, your exposure time with a 10 inch scope is only one quarter of that for a 5 inch scope.

But what is this business about resolution? As you may remember, the physical nature of light is designed to confuse the best trained physicist, let alone the poor layman. Sometimes light behaves like particles—like photons, little packets of energy; and other times it behaves like waves on a pond. Actually, light always behaves like light. It is just the way we interpret it that changes. For the purposes of our discussion we can ignore the particle business and consider light to be a simple wave.

If light is a wave, then it has a wavelength—the distance between peaks, which is called L . Consider two sine waves, W and W' , with the same wavelength.

Let W' be delayed with respect to W by an amount ϕ (phi, pronounced “fee”). Add the two waves together and the results depend on ϕ . If ϕ is zero, the peaks fall on top of each other and the two waves reinforce each other. But if ϕ is $L/2$ then the peaks of W fall in the troughs of W' and vice versa. In this case, the waves are




Fig. 1-14 Huygens' wavelets and the geometry of the slit.

out of phase and cancel each other out leaving darkness. This is called destructive interference. (See Figure 1-13.)

Christiaan Huygens noted that if you have an opening in a breakwater and a train of parallel waves approaching the gap, as the waves pass through they will radiate out from the gap appearing more or less circular. (See Figure 1-14.)

You can easily demonstrate this in a quiet swimming pool with a couple of kick boards. Actually each point on the advancing wave front radiates circular waves, but adjacent points cancel the sidewise parts of the circle and result in a straight wavefront.

Now consider the wavelets radiating from the points *A* and *B*—the edges of the gap. As they move off perpendicular to the gap (straight down in figure), the wavelets from *A* and *B* are in phase, so when they strike the back wall they add up and we get a rough spot. Now consider the wavelets moving off slightly to the right at an angle *X* to the perpendicular. We have the tangent of *X* equal to *C* over *D*:

$$\tan X = \frac{C}{D},$$

and if *C* is just $L/2$ these waves arrive at the back wall exactly out of phase and cancel each other out. We have a calm spot. When *X* is such that *C* is equal to $3L/2$, or $5L/2$ and so on we have further calm spots, and when *C* is equal to $n \cdot L$ we have spots where the waves add up and the surf is rough. The points between *A* and *B* also contribute to the results, and if you do the math you get waves whose heights look something like Figure 1-15.

So if you have a point source someplace way out in the ocean radiating waves, they are very nearly straight and parallel by the time they reach the break-




Fig. 1-15 Distribution of light intensity in a diffraction pattern.

water, and you will get a pattern whose strength looks like the curve in Figure 1-15. The width of the central peak depends on the angle X that causes the first null. For small angles, measured in radians,

$$X = \tan X$$

so

$$X = \frac{L}{2}D$$

where L is the wavelength and D the width of the opening. This is just an approximation, but it is close enough for government work. If there are two widely separated sources out there you will get two overlapping patterns with the separate peaks readily distinguishable. But let the sources approach each other. Eventually the peaks on the back wall will overlap and you won't be able to tell if there is one source or two. The minimum angular difference between the two sources at which you can still tell that there are two sources is called the *angular resolving power*. As is true with water waves, so also is true with light waves. If two stars have too small an angular separation, their diffraction disks will overlap and the observer won't be able to tell if there is one star or two. This is called the resolving power of the instrument. Going from linear to circular and doing the integrals properly we have:

$$X = \frac{1.17}{R}$$

where X is measured in arc minutes, R is the radius of the objective in centimeters, and we are dealing with green light whose wavelength is 5.5×10^{-5} cm.

Device	Radius	Wavelength	Resolving power
Human eye	0.1 cm	$5 \cdot 10^{-5}$ cm	0.42 arc min
6-inch scope	7.6 cm	$5 \cdot 10^{-5}$ cm	0.16 arc min
Mt. Wilson 100-inch	127 cm	$5 \cdot 10^{-5}$ cm	0.56 arc sec
Keck 8 meter	400 cm	$5 \cdot 10^{-5}$ cm	0.18 arc sec
35 foot radio scope	533 cm	20 cm	15 arc deg
Coast to coast radio interferometer	$2.5 \cdot 10^8$ cm	20 cm	0.056 arc sec

For radar waves the wavelength is around 20 cm or 4×10^5 times as long as green light. The resolving power of an optical device is inversely proportional to the wavelength, so radio telescopes have to be 400,000 times as big as optical scopes to achieve the same resolution.

The human eye in daylight has an iris opening of about 2 millimeters, which should give a resolution of 0.42 arc minutes. But the eye can only resolve about one arc minute. It is interesting to note that evolution has adjusted the size of the rods and cones in the retina so that their size very nearly matches the angular resolution of the pupil. Some typical angular resolutions are given in Table 1-3.

1.11 The Doppler Effect

Christian Johann Doppler was born in Salzburg, Austria on November 29, 1803 and died in Venice on March 17, 1853. His birthplace is marked with a plaque declaring it to be the “Gebursthaus” of the famous physicist. The brief discussion of the Doppler effect in my encyclopedia is, if not in error, very misleading. What a wonderful opportunity for a tutorial! We former teachers will stop at nothing...

—CCF

The Doppler effect has to do with whether the distance between you and a sound source is decreasing or increasing; i.e., whether the sound source is approaching or receding. Exactly the same effect is observed with light rays, so it is of interest to astronomers who want to measure the radial velocity of distant stars. We will use sound waves to explain things because they are easier (for me) to understand.

Sound waves are characterized by having a “pitch.” This is the number of vibrations per second striking a listener’s ears. If you have ever looked at the keyboard of a large pipe organ you will have noticed stops marked “8 ft.” and “16 ft.” and perhaps even “32 ft.” These numbers refer to the length of the pipes, so they also refer to the wavelength of the sounds the pipes produce.

Given still air at sea level pressure at some fixed temperature and fixed humidity, sound travels with a characteristic velocity known as, you guessed it, the “speed of sound.” We will designate this by the letter C . Consider a note of 250 cps (cycles per second)—one just above a musician’s middle A. Successive points




Fig. 1-16 Diagram of the Doppler effect, with a stationary auditor and a moving car horn.

of maximum pressure are separate in time by $1/250$ second, or 4 milliseconds. Since C is close to 1000 feet per second, the previous pressure peak will have moved off by 4 feet by the time the next pressure peak comes along. Thus the distance between successive peaks, or the wavelength, is 4 feet. Using L for wavelength and putting the last couple of sentences into equations,

$$fL = C$$

frequency times wavelength equals the velocity of propagation.

Consider a car far away blowing its horn. Suppose the car is standing still and there is no wind to complicate life. Let the sound of the horn be a pure tone of 1000 vibrations per second. The speed of sound at sea level is close to 1000 feet per second. That's why a count of 5 between the lightning flash and the thunder indicates a distance of about 5000 feet, or one mile. There will be $1/1000$ of a second between peaks of the sound waves, and therefore they will be spaced by 1 foot.

Now let the car be moving at 100 feet per second toward the auditor. That's just about 70 miles per hour and easily attainable. Let the speed of the car stay constant, and let the car be moving directly toward the auditor. (There may be some difficulty after a while in finding volunteer auditors for our experiments, but nobody ever said physics was going to be easy...) During each millisecond, the car will move 0.1 foot toward the auditor. Therefore, when the horn emits its next peak, the previous peak will be only 0.9 feet ahead of the new one. The car has caught up, in part, to the previously emitted sound. The frequency of a sound wave is the number of peaks that arrive in a second. Since the peaks are only 0.9 feet apart and they advance at 1000 feet per second, the number arriving is 1111 per second. That makes a note a little bit higher than that of the stationary horn.

32 Chapter 1: Matters Mathematical

If the car is moving away from the auditor the next sound wave will be emitted after the car has moved off 0.1 foot so the peaks of the sound waves will be 1.1 feet apart. The auditor will hear a note of 909 cycles per second—a little lower than the stationary horn. In neither case does the frequency of the note change as long as the car is moving with a constant velocity. Nor is there any minimum speed below which the effect doesn't take place. The change in pitch may be too subtle to notice, but it is always there. Let the car be aimed to pass a few feet to one side of the auditor. At once it becomes easier to recruit volunteers and we introduce a complicating factor in the experiment. As the car moves past the auditor, at the exact moment when it is even with the listener, it is neither approaching nor receding from him. It was coming toward him at 70 miles per hour (almost) and in a second or two it will be moving away at the same speed, but just as it passes, its distance is constant. Look at **Figure 1-16**.

$$D_1 = \sqrt{X_1^2 + H^2}$$

$$D_2 = \sqrt{X_2^2 + H^2}$$

and

$$X_2 = X_1 - vt.$$

In this case we get what a barbershop quartet calls a “swipe.” The sound starts on one note and then quickly descends to a second lower note. The closer you stand to the car's path, the quicker the swipe is over. Of course, if you stand too close, you won't hear the swipe at all.

Now see what the Britannica has to say (note that they are moving the auditor rather than the source):

...when the blowing horn is passed at any speed above 10 m.p.h. the pitch of the note becomes increasingly lower.

I don't know where the 10 m.p.h. gets into the act. Does that lead to a just noticeable difference in pitch? But it reads as if the frequency will continue to drop forever...?